BYTE®

the small systems journal

JANUARY 1981 Vol. 6, No. 1
$2.50 in USA/$2.95 in Canada
A McGraw-Hill Publication

HAND-HELD COMPUTERS

•The HHC retains the contents of memory even when it is turned off. In addition, you do not lose what you are working on if you accidentally hit the OFF button. These are important features that indicate the amount and depth of human engineering that has been applied to the design of the HHC.

•The HHC will be marketed aggressively by both Quasar and Panasonic. The public reaction to this device, which is the first of its kind to be marketed on such a large scale, will be carefully observed by manufacturers and may determine the extent and direction of future consumer products in this area. We feel that the Panasonic/Quasar HHC is highly qualified to receive this scrutiny and that the public response will be favorable.■

Acknowledgment

The cover photograph and all interior photographs are by Ed Crabtree. Photo 2 is courtesy Quasar Electronics Company.

Another Pocket Computer

The internal architecture of the TRS-80 Pocket Computer is radically different from the other pocket computers now reaching the market. Instead of a single 8-bit microprocessor (such as that used in the Quasar/Panasonic HHC and the Sinclair ZX-80), the designers of the TRS-80 Pocket Computer (Sharp Electronics of Japan) decided to use two 4-bit microprocessors in a unique serial configuration.

Both microprocessors are custom CMOS (complementary metal-oxide semiconductor) integrated circuits with built-in ROM (read-only memory). The purpose of microprocessor 1 is to arrange data and make decisions. It reads the data that is keyed in or fetched from programmable memory. It is also responsible for parsing arithmetic operations and interpreting the syntax of BASIC statements. It then arranges the data and provides instruction codes to microprocessor 2 through a transfer buffer. The actual execution of an instruction is performed by microprocessor 2, which also updates the display and notifies microprocessor 1 that it has finished its function. The respective duties of the microprocessors are listed at right.



Microprocessor 1	Microprocessor 2
Key input routine	*Display processing routine* *Input buffer* *Computational result* *Error*
Acknowledgment of the remaining program	*Arithmetic routine*
One instruction to one program step incorporation	*Character generator*
Interpreter: *Program execute statement* *Cassette control statement* *Command statement* *Printer control (reserved)*	*Cassette routine* *Print routine* *Buzzer*
Execution of manual operation	*Recognition of printer (reserved)*
Power shut-off control	*Power off*
Clock stop control	*Clock stop*

Memory Organization

The programmable memory of the TRS-80 Pocket Computer is contained in four integrated circuits. There are three memory ICs, each containing 512 bytes of programmable memory. The three ICs which drive the liquid-crystal display each contain 128 bytes of programmable memory. Putting it all together, you end up with 1920 bytes of programmable memory. After you subtract memory space used for the transfer buffer, input buffer, display buffer, fixed memories, and reserved keys, you end up with 1424 bytes of user-addressable memory. Into this space you can easily fit a BASIC program of around 250 lines (average length)...**SM**■

Numerical Analysis for the TRS-80 Pocket Computer

Mike Salem, 26A Delancey St, London NW1 7NH, England

Complicated programs can often be easily modified to fit into the new pocket computers. I've taken three programs from the December 1979 issue of BYTE and modified them to run on the Radio Shack TRS-80 Pocket Computer (sold as the Sharp PC-1211 outside of the United States). The Pocket Computer has a 24-character LCD (liquid-crystal display), twenty-six fixed variables, and 1424 bytes of programmable memory.

One of the programs I modified was the discrete-Fourier-transform program that appeared in "Frequency Analysis of Data Using a Microcomputer" by F R Ruckdeschel (December 1979 BYTE, page 10). I also combined two programs that compute the time-domain response of a system with a given transfer function into a single program ("Noniterative Digital Solution of Linear Transfer Functions" by Brian Finlay, December 1979 BYTE, page 144). The modified programs have all of the features of the originals, with the obvious omissions of printing and plotting.

Incidentally, it is important to note that the TRS-80 Pocket Computer, in common with many machines, allows BASIC lines to contain multiple statements (saving 3 bytes of programmable memory for each line number omitted). Although this feature is useful in itself, the TRS-80 Pocket Computer also has an IF statement that can control all of the remaining statements in the

Listing 1: *A discrete-Fourier-transform program for the TRS-80 Pocket Computer. This program was modified from "Frequency Analysis of Data Using a Microcomputer" by F R Ruckdeschel (December 1979 BYTE, page 10). Statements entered on the same line are separated here for clarity.*

```
 10   :REM BYTE DEC 79
 11   :RADIAN
190   :INPUT "1ST X? ";Z,"LAST X? ";Y,"#OF POINTS? ";N
250   :I=1
      :INPUT "I/P SCALE FACTOR? ";I
      :IF I<1 GOTO 250
290   :D=(Y-Z)/(N-1)
      :Q=0
      :V=π/DI
      :U=V/(N-1)
340   :FOR I=1 TO N
      :PAUSE "NEXT # = ";I
      :BEEP 1
      :INPUT "NEXT F(T) VALUE? ";O
      :A(I+26)=O
      :NEXT I
370   :B=0
      :FOR I=27 TO N+26
      :IF B>A(I) LET B=A(I)
410   :NEXT I
420   :    FOR I= 27 TO N+26
      :    A(I)=A(I)-B
      :    NEXT I
      :B=ABS B
      :T=0
      :FOR I=27 TO N+26
      :IF T<A(I) LET T=A(I)
510   :NEXT I
710   :FOR I=1 TO N
      :W=(I-1)*U
      :C=0
      :P=0
      :    FOR M=1 TO N
      :    X=Z+(M-1)*D
      :    G=WX
770   :    C=C+A(M+26)*COS G
      :    P=P+A(M+26)*SIN G
      :    NEXT M
800   :F=√(PP+CC)*D
      :IF I=1 LET C=C-NB
      :         F=D*ABS C
810   :BEEP 1
      :PRINT U*(I-1);"RAD/S"
      :PRINT "AMPL. =";F
815   :IF C<>0 LET O=ATN(P/C)*180/π
      :              PRINT "PHASE=";O
820   :NEXT I
      :BEEP 3
      :PRINT "END OF RUN"
      :END
```

same line. Since this makes listings a bit difficult to read, I prepared listings 1 and 2 with a separate statement on each line. ■

Listing 2: *A program for the TRS-80 Pocket Computer that computes the time-domain response of a system with a given transfer function. The program shown was combined and modified from two programs contained in "Noniterative Digital Solution of Linear Transfer Functions" by Bryan Finlay (December 1979 BYTE, page 144).*

```
10    :REM "TF: TRANSFER FCN - BYTE DEC 79"
70    :RADIAN
      :INPUT "CONST.? ";K,"#TERMS NUM.? ";E,"#TERMS
      DEN.?";L
150   :IF E=0 GOTO 240
160   :    FOR G=27 TO E+26
      :    O=10+G
      :    INPUT "RL, NUM.? ";A(G),"IM, NUM.? ";A(O)
      :    NEXT G
240   :IF L=0 GOTO 330
250   :    FOR H=47 TO L+46
      :    O=10+H
      :    INPUT "RL, DEN.? ";A(H),"IM, DEN.? ";A(O)
      :    NEXT H
330   :FOR G=1 TO L
      :O=66+G
      :Q=76+G
      :A(O)=1
      :A(Q)=0
      :IF E=0 GOTO 450
370   :    FOR H=1 TO E
      :    D=A(26+H)-A(46+G)
      :    C=A(36+H)-A(56+G)
380   :    M=√(DD+CC)
      :    N=ATN(C/D)
      :    IF D<0 LET N=N-π
410   :    A(O)=MA(O)
      :    A(Q)=N+A(Q)
      :    NEXT H
450   :FOR R=1 TO L
      :IF R=G GOTO 501
465   :D=A(46+R)-A(46+G)
      :C=A(56+R)-A(56+G)
470   :M=√(DD+CC)
      :N=ATN(C/D)
      :IF D<0 LET N=N-π
500   :A(66+G)=A(66+G)/M
      :A(76+G)=A(76+G)-N
501   :NEXT R
      :NEXT G

520   :INPUT "T(0)? ";O,"DT? ";S,"# STEPS? ";N
      :T=O+NS
620   :U=-S
      :FOR Q=1 TO N
      :U=U+S
      :V=0
      :W=0
      :H=1+INT((U-O)/S)
650   :    FOR G=1 TO L
      :    X=A(66+G)*EXP(-UA(46+G))
      :    Y=A(76+G)-UA(56+G)
      :    V=V+X*COS Y
      :    W=W+X*SIN Y
      :    NEXT G
710   :Z=K*√(VV+WW)*SGN V
      :BEEP 1
      :PRINT "TIME=";U
      :PRINT "RESP.=";Z
730   :NEXT Q
```